Examples for Mn in Steel – spectrophotometric method

Values in red are measured or recorded numbers. Values in blue are calculated.

Preparing a standard Mn solution

82.7 mg of Mn dissolved in acid and diluted to 1.00 L

1 ppm = 1 mg L^{-1} so the stock solution is 82.7 ppm Mn

Preparing the Steel Unknown

1.0032 g of steel dissolved in acid and diluted to 250 mL

Standard Addition Solutions

Solutions were prepared according to the following table, diluted to 50 mL and absorbance values are reported.

Sample	Steel (mL)	H_3PO_4 (mL)	Std. Mn Solution (mL)	KIO_4 (g)	Absorbance
1 (blank)	20.0	5.0	0	0.0	--
2	20.0	5.0	0	0.4	0.0786
3	20.0	5.0	1.0	0.4	0.1540
4	20.0	5.0	2.0	0.4	0.2264
5	20.0	5.0	3.0	0.4	0.3004
6	20.0	5.0	4.0	0.4	0.3785
7	20.0	5.0	5.0	0.4	0.4483

Calculate the concentration of <u>added</u> Mn in each sample
For example, sample number 4 has 2.0 mL of added Mn in a 50 mL total volume

82.7 ppm x (2 mL / 50 mL) = 3.3080 ppm Mn

Note: Your solution number 2 should have an absorbance of ~0.1 , adjust your steel sample volume accordingly. For example if your number 2 solution has an absorbance of 0.18 using 20 mL, reduce the volume to 10 mL which should give you an absorbance of 0.09. Don't forget to use 10 mL instead of 20 in the following calculations.

Therefore we get the following data…

Sample	Mn (ppm)	Absorbance
1 (blank)	--	--
2	0.0000	0.0786
3	1.6540	0.1540
4	3.3080	0.2264
5	4.9620	0.3004
6	6.6160	0.3785
7	8.2700	0.4483

The resulting plot of Mn (ppm) vs. Absorbance is shown below.



Linear least squares regression analysis gives a slope = 0.04484 and the y – intercept = 0.7894

In the graphical method of standard additions the concentration of Mn from the steel sample in the measured solution is equal to the absolute value of the x – intercept.

The x – intercept is equal to the negative y – intercept divided by the slope

= (-0.7894/0.04484) = -1.7603 ppm

Therefore the Mn concentration in the measured sample from steel is equal to:

1.7603 ppm (mg L^{-1})

Now we need to know how much Mn was in the total steel sample.

Recall that the steel was digested then diluted to 250 mL. We then took 20 mL of this solution and diluted to 50 mL for the absorbance measurement.

1.7656 mg L^{-1} Mn x (0.050 L / 0.02 L) x 0.250 L = 1.1035 mg of Mn total

Finally we can calculate the mass percent of Mn in our steel sample

(1.1035 x 10^{-3} g Mn) / (1.0032 g steel) x 100 = 0.11 %

We must also calculate the confidence interval for this method. In order to do this we need to know the standard deviation of the x – intercept (s_x). The equation for this is given as 1.12 in the introduction of your lab manual. Once you have calculated this number, you will use equation 1.10 to calculate your confidence interval.

(equation 1.10) $\mu = x \pm ts_x$

Note: this is not the same as you use in other methods and that the t value is taken as n – 2 degrees of freedom (where n is the number of points on your plot ; 6 in the above example for 4 degrees of freedom).

There are a lot of calculations for this experiment and it is much easier to let a spreadsheet do the work. Create a worksheet in excel similar to the one that follows these notes. You will need to enter all of the values listed with light blue backgrounds including the t table in cells D15 to E25.

All of the formulas listed below should be preceded with and equal sign (=) so that Excel knows what follows in an equation and not just text.

Equations for cells **G2** through **L2** are given. Copy these cells and paste them into cells **G3** through **L8** (this will fill the upper right green cells)

The added Mn concentration is calculated in cell **D2**, copy and paste this into **D3** to **D8**

Cell **B5** should be **=B1*1000**

Cell **D9** is the sum of the above numbers and should be copied into cells **E9** and **G9** through **L9**

The equations in cells **B9** to **B23** will be used to calculate the wt% of Mn in your sample as well as the correct 95% confidence limit.

It is worthwhile to set this spreadsheet up because you can save it and use it in the future as a tool when performing quantitative analysis. We will also be modifying this spreadsheet for calculating weight percents of copper and lead in the brass analysis next.

Important:

To check the accuracy of your spreadsheet enter the numbers in red in the example sheet which follows and you should get a wt% of 0.11 and 95% CI of 0.007. If you don't then compare all of the values in the calculated cells to see which one doesn't match.

CELL	VALUE	FORMULA
B9	slope	SLOPE(E2:E7,D2:D7)
B10	intercept	INTERCEPT(E2:E7,D2:D7)
B11	sy	SQRT(L9/(B12-2))
B12	n	COUNT(E2:E7)
B13	sqrt D	SQRT((J9*B12)-(G9*G9))
B14		
B15	x-intercept	-B10/B9
B16	sd x-int.	(B11/B9/B13)*SQRT((B12*B15^2)-(2*B15*G9)+J9)
B17	RSD x-int	-B16/B15
B18		
B19	mass of Mn	-B15*(50/B3)*0.25/1000
B20		
B21	**wt%**	B19/B2*100
B22	sd wt%	B21*B17
B23	**95% CI**	(B21*B17)*LOOKUP(B12-2,D16:E25)
D2	[Mn]	$B$5*C2/$B$4
D9	sum x	SUM(D2:D7)
G2	x	D2
H2	y	E2
I2	xy	D2*E2
J2	x^2	D2^2
K2	di	E2-(($B$9*D2)+$B$10)
L2	di^2	K2^2

	A	B	C	D	E	F	G	H	I	J	K	L
1	**Mn (g)**	**0.0827**	**vol Std**	**[Mn]**	**Abs 524**		**x**	**y**	**xy**	**x^2**	**di**	**di^2**
2	**Steel**	**1.0032**	**0**	0.0000	**0.0786**		0	0.0786	0	0	-0.000338	1.143E-07
3	**aliquot size**	**20**	**1**	1.6540	**0.1540**		1.654	0.154	0.254716	2.735716	0.0008905	7.929E-07
4	**total Vol (50)**	**50**	**2**	3.3080	**0.2264**		3.308	0.2264	0.7489312	10.942864	-0.000881	7.761E-07
5	[Mn] - ppm	**82.700**	**3**	4.9620	**0.3004**		4.962	0.3004	1.4905848	24.621444	-0.001052	1.108E-06
6			**4**	6.6160	**0.3785**		6.616	0.3785	2.504156	43.771456	0.0028762	8.272E-06
7			**5**	8.2700	**0.4483**		8.27	0.4483	3.707441	68.3929	-0.001495	2.236E-06
8												
9	slope	0.04484	SUMS=>	24.81	1.5862		24.81	1.5862	8.705829	150.46438	2.22E-16	1.33E-05
10	intercept	0.07894										
11	sy	1.82E-03										
12	n	6										
13	sqrt D	16.9485										
14												
15	x-intercept	-1.7603		**DOF**	**t (95%)**							
16	sd x-int.	0.0384		1	12.706							
17	RSD x-int	2.18%		2	4.303							
18				3	3.182							
19	mass of Mn	**0.00110**		4	2.776							
20				5	2.571							
21	**wt%**	**0.11**		6	2.447							
22	sd wt%	0.0024		7	2.365							
23	**95% CI**	**0.007**		8	2.306							
24				9	2.262							
25				10	2.228							
26